NASA Contractor Report 178241

CALCULATION OF SIDEWALL BOUNDARY-LAYER PARAMETERS FROM RAKE MEASUREMENTS FOR THE LANGLEY 0.3-METER TRANSONIC CRYOGENIC TUNNEL

A. V. Murthy

Vigyan Research Associates, Incorporated
Hampton, Virginia

contract NAS1-17919

FEBRUARY 1987

(NASA-CR-178241) CALCULATION OF SIDEWALL BOUNDARY-LAYER PARAMETERS FROM RAKE MEASUREMENTS FOR THE LANGLEY 0.3-METER TRANSONIC CRYOGENIC TUNNEL (Vigyan Research Associates, Inc.) 36 p CSCL 01A G3/02

N87-16807

Unclas
44006

National Aeronautics and
Space Administration

Langley Research Center
Hampton, Virginia 23665

Summary

Correction of airfoil data for sidewall boundary-layer effects requires a knowledge of the boundary-layer displacement thickness and the shape factor with the tunnel empty. To facilitate calculation of these quantities under various test conditions for the Langley 0.3-m Transonic Cryogenic Tunnel, a computer program has been written. This program reads the various tunnel parameters and the boundary-layer rake total head pressure measurements directly from the Engineering Unit tapes to calculate the required sidewall boundary-layer parameters. Details of the method along with the results for a sample case are presented.

INTRODUCTION

The presence of boundary-layers on the sidewalls of a two-dimensional wind tunnel are known to affect the test data on the airfoils. Recently, methods have been proposed to account for this sidewall boundary-layer interference. A summary of these correction methods may be found in reference (1). For application of these methods, it is necessary to have a knowledge of the values of the empty test-section sidewall boundary-layer displacement thickness and the shape factor. These values are generally deduced from the total head pressure measurements across the boundary-layer, during the calibration phase of the tunnel.

For the Langley 0.3-m Transonic Cryogenic Tunnel, the sidewall boundary-layer measurements were recently made using a wall mounted boundary-layer rake in the newly installed adaptive wall test-section. To calculate the boundary-layer displacement thickness and the shape factor from these measurements, a Fortran computer program was written. The present report gives a brief description of the program developed and its use for calculating

the required sidewall boundary-layer characteristics directly from the tunnel data stored on the Engineering Unit tapes.

Nomenclature

b	:	Test-section width
H	:	Boundary-layer shape factor
L	:	Length scale
M	:	Mach number
n	:	Index in power-law for the velocity profile ($U \propto y^{1/n}$)
P	:	Total pressure
Pr	:	Prandtl number
p	:	Static pressure
r	:	Boundary-layer recovery factor
T	:	Static temperature
U	:	Velocity
y	:	Coordinate normal to the sidewall
γ	:	Ratio of specific heats
δ	:	Boundary-layer thickness
δ^*	:	Boundary-layer displacement thickness
θ	:	Boundary-layer momentum thickness

Subscripts

e	:	refers to conditions outside the boundary-layer
bl	:	refers to boundary-layer
∞	:	refers to freestream conditions

Description of the Method

The details of calculation of the boundary-layer displacement thickness and the shape factor are given in reference (2). First, the measured total head pressures in the boundary-layer are converted to Mach numbers using the isentropic relation, and assuming the static pressure across the boundary-layer to be

constant at the freestream value. The isentropic relation for the Mach number is given by (for $\gamma = 1.4$)

$$M_{bl} = [5\{(P_{bl}/p)^{.2857} - 1\}]^{0.5} \tag{1}$$

Knowing the Mach number across the boundary-layer, the velocity variation can be calculated using the relation

$$(U_{bl}/U_e) = (T_{bl}/T_e)^{1/2}(M_{bl}/M_e) \tag{2}$$

The temperature measurements are generally not made, and hence it is necessary to assume a suitable variation acrosss the boundary-layer. For the adiabatic wall, the relation between the velocity and temperature profiles can be expressed by the Crocco's relation (reference 3) given by (for $\gamma = 1.4$),

$$(T_{bl}/T_e) = 1 + 0.2rM_e^2[1 - (U_{bl}/U_e)^2] \tag{3}$$

where r is boundary-layer recovery factor. The value of r depends on the nature of the boundary-layer, and the value of the Prandtl number (Pr). For the turbulent boundary-layer, this value is approximately $Pr^{1/3}$. Using Crocco's relation, the velocity and temperature variation across the boundary-layer can be calculated. The boundary-layer displacement thickness, momentum thickness and the shape factor are then calculated by integrating the expressions:

$$\delta^* = \int_0^\delta [1 - (T_e/T_{bl})(U_{bl}/U_e)]\, dy \tag{4}$$

$$\theta = \int_0^\delta (T_e/T_{bl})(U_{bl}/U_e)[1 - (U_{bl}/U_e)]\, dy \tag{5}$$

and

$$H = (\delta^*/\theta) \tag{6}$$

For calculating the incompressible values of the displacement and momentum thickness, the factor (T_e/T_{bl}) is set equal to unity, in equations (4) and (5).

Program Description

The method described above has been incorporated into a Fortran computer program called BOUNLAY. The program reads the required input data either from the Engineering Unit tapes or from user input. The boundary-layer rake tubes are connected to scanivalve. The values of the tunnel stagnation pressure, stagnation temperature and the static pressure are read at each instant the boundary-layer pressure is measured. These instantaneous values are used in the calculation of the boundary-layer parameters.

The boundary-layer rake used in the 0.3-m TCT (figure 1) has fifteen tubes and the total height is about 0.6". The tube closest to the wall is about 0.024" from the wall. Hence, to evaluate the boundary-layer thickness, it is necessary to extrapolate the velocity profile up to the wall in a suitable manner. Also, under certain flow conditions, it is probable that the outermost tube may not be positioned within the boundary-layer edge. This requires suitable extrapolation of the measured data from the outermost tube to the outer edge of the boundary-layer. Hence, a power-law variation was assumed for the turbulent boundary-layer velocity profile. The velocity profile was represented by the equation

$$(U_{bl}/U_e) = (y/\delta)^{1/n} \tag{6}$$

or

$$\log(U_{bl}/U_e) = (1/n)\log(y/L) + (1/n)\log(L/\delta) \tag{7}$$

where 'δ' is the boundary-layer thickness, 'n' is the index of the power-law variation and L is a length scale. The values of 'δ' and 'n' are determined by making a least square fit to the calculated values of (U_{bl}/U_e) at different heights across the boundary-layer. The least square fit is accomplished by using the library subroutine LSQPOL. An example of this least square fit result is shown in figure 2. The slope and the intercept

of the least square fit determine the values of 'n' and 'δ' respectively.

After determining the index of power-law variation for the velocity, the integrals for the displacement thickness and the momentum thickness are evaluated by using the library subroutine CADRE.

It was recognised that the readings of some of the tubes may be erroneous due to extraneous effects during tests. These measurement points need to be omitted to avoid erroneous calculation of the sidewall boundary-layer parameters. Hence, provision has been made to skip up to three tube locations.

Program Input

The input data to the program BOUNLAY is through the namelist PARAM and the relevant parameters with their default values are given in Appendix A. The value of the parameter IDATA, either '0' or '1', specifies whether the required tunnel parameters are to be read from the Engineering Unit tapes, or are given directly as input by the user. The tube heights from the wall are specified through the parameter Y. The default values shown correspond to the recent measured data, and can be changed through the namelist, if new measurements are made, or if a new rake is used. Corresponding to the default option of IDATA=0, if the tunnel parameters are to be read from the Engineering Unit tapes, the values of the array numbers where the various parameters are stored are to be specified through the namelist. These parameters and other control parameters required are indicated in Appendix A.

A typical set of input data along with the control cards required are shown in Appendix B. Following the namelist PARAM, the test, run and point numbers are specified in 3I3 format, when the tunnel parameters are to be read from the Engineering Unit

tapes. Specifying '0', for the point number allows the program to calculate the sidewall boundary-layer parameters for all the points in the specified run number. For termination of the calculation, the run number is specified as '999', in the last record.

If the tunnel parameters are given directly as input data, following the namelist PARAM, the program reads a title card in the format 10A8. After this, the values of the stagnation pressure(psia), stagnation temperature(degree K), height of the tube from the wall(inches), boundary-layer pressures(psia) and tunnel static pressure (psia) are specified in format 5F10.6. These are repeated corresponding to the value of NBL specified in the namelist PARAM.

Results for a Sample Case

The use of the computer program developed is demonstrated by applying it to one of the runs conducted recently during empty test section calibration (Test 200). The sample output shown in Appendix C, corresponds to the results obtained with the data in Appendix B. The corresponding nominal freestream Mach number and Reynolds number are 0.7 and 30 million/ft, respectively (Run number 19 and Point number 6).

First, the values of the various parameters in namelist PARAM are printed. Following this, the title, and the test, run and point numbers are printed. Subsequently, the tunnel conditions averaged over the time interval of the boundary-layer measurements, the calculated values of the boundary-layer parameters are printed. At the end, the variation of the pressure, temperature, velocity and Mach number across the boundary-layer are tabulated.

If the value of the parameter IPLOT is not equal to zero, the calculated velocity profiles are written on a temporary file

on Tape 10. These are then plotted using the program UPLOT, and the corresponding control cards required for plotting are shown in Appendix A. The program UPLOT uses the library subroutine INFOPLT and plots both the measured velocities and the calculated power-law profile for comparison (see Figure 3).

For the sample case considered, it is seen that the value of the displacement thickness and the shape factor at the model location are .0713" and 1.4595 respectively. For the testsection width of 13", the value of the parameter ($2\delta^*/b$) which is one of the parameters characterizing the extent of sidewall boundary-layer correction is about 0.01. The corresponding corrrection for the test Mach number on long chord models is about -0.008. However, for short chord models, it is likely the sidewall boundary-layer correction will be much smaller. The detailed calculation of the sidewall boundary-layer parameters, and the corresponding corrections over the operating range of the 0.3-m TCT will be dealt in a future report.

References

1. Murthy, A. V.: A Simplified Fourwall Interference Assessment Procedure for Airfoil Data Obtained in the Langley 0.3-Meter Transonic Cryogenic Tunnel. (NASA CR under publication)

2. Murthy, A. V.; Johnson, C. B.; Ray, E. J.; Lawing, P. L.; and Thibodeaux, J, J.: Studies of Sidewall Boundary-Layer in the Langley 0.3-Meter Transonic Cryogenic Tunnel With and Without Suction. NASA TP-2096, March 1983.

3. White, Frank, M.: Viscous Fluid Flow. McGraw-Hill, Inc., c.1974.

Appendix A

Parameters in Namelist PARAM

IDATA : = 0, Tunnel parameters are read from EU tapes, (0)*
= 1, Tunnel parameters are given as input data

ICOMP : = 0, Incompressible values of δ^* and θ calculated, (1)
= 1, Compressible values of δ^* and θ calculated

IPLOT : = 0, Plot files are not generated, (1)
= 1, Plot files are generated

L : Reference length scale, (1.0)

NBL : Number of tubes in the boundary-layer rake,
Maximum value = 30, (15)

Y : Tube heights from the wall (.024,.06,.10,.14,.18,
.22,.26,.30,.34,.38,.42,.46,.50,.55,.59)

ISKIP : Number of tubes to be omitted calculation, Max. 3 (1)

IBL : Port number of the tubes to be omitted (1, 0, 0)

IPRINT : = 0, Only final results are printed (0)
= 1, Intermediate results also printed

IOPT : = 1, δ^* and θ calculated using the power-law for the entire profile.
= 2, δ^* and θ calculated using the power-law only upto first tube, and using trapezoidal rule for the remaining profile

NTEST : Array number for the test+ (3)

NRUN : Array number for the run+ (5)

NPOINT : Array number for the point+ (6)

NPORT : Array number for the port+ (10)

NBLSCAN : Array number for the boundary-layer rake pressure+ (117)

NPTINF : Array number for the tunnel total pressure+ (107)

NPRT : Array number for the tinnel total temperature+ (217)

NPTC : Array number for the tunnel static pressure+ (108)

* Numbers within paranthesis refer to default values

+ to be specified only when IDATA = 0.

Appendix B

Control Cards and Sample Input Data

```
PROBOUN,T1000,CM170000.                    Bldgnum    name
USER,usernum.passwor.
CHARGE,chgnum,LRC.
ATTACH,TAPE7=D*******/UN=usernum.
GET,BOUNLAY.
FTN,I=BOUNLAY,L=0,PL=10000.
ATTACH(FTNMLIB/UN=LIBRARY,NA)
LDSET(LIB=FTNMLIB,PRESETA=INDEF)
MAP,OFF.
LGO.
REWIND,TAPE10.
RETURN(*,TAPE10)
GET,UPLOT
FTN,I=UPLOT,L=0.
ATTACH,LRCGOSF/UN=LIBRARY,NA.
LDSET(LIB=LRCGOSF,PRESETA=INDEF)
LGO.
DELIVER.bldgnum  name
PLOT.VARIAN,FF
/eor
 $PARAM IDATA=0,IOPT=2,$END
200019000
200999000
 $PARAM IDATA=1,IOPT=2,$END
 TEST NO:  200.   RUN:  19.    POINT:  6.   SAMPLE INPUT DATA
 35.5818  129.6503    .0240   29.1408   25.6570
 35.5475  129.8012    .0600   30.9015   25.6787
 35.5341  129.8498    .1000   31.5694   25.5685
 35.6083  129.8417    .1400   31.9734   25.4811
 35.6049  129.8197    .1800   32.5464   25.5275
```

continued

35.6106	129.7544	.2200	32.8914	25.5226
35.6429	129.6369	.2600	33.5220	25.5598
35.6908	129.5605	.3000	33.8837	25.6194
35.5991	129.6841	.3400	34.2634	25.6547
35.5794	129.7295	.3800	34.8155	25.5787
35.5810	129.7403	.4200	34.8841	25.5163
35.6308	129.7172	.4600	35.1331	25.5355
35.5785	129.7094	.5000	35.2827	25.5193
35.6037	129.6764	.5500	35.4686	25.5294
35.6374	129.5828	.5900	35.4827	25.4787

/eor

/eof

Appendix C
Sample Output from the Program

```
$PARAM
NTEST   = 3,
NRUN    = 5,
NPOINT  = 6,
NBL     = 15,
NPORT   = 10,
NBLSCAN = 117,
NPTINF  = 107,
NPTC    = 108,
NPRT    = 217,
ICOMP   = 1,
IPLOT   = 1,
Y       = .24E-01, .6E-01, .1E+00, .14E+00, .18E+00,
          .22E+00, .26E+00, .3E+00,  .34E+00, .38E+00,
          .42E+00, .46E+00, .5E+00, .55E+00, .59E+00,
          I, I, I, I, I, I, I, I, I, I, I, I, I, I, I,
L       = .1E+01,
ISKIP   = 1,
IBL     = 1, 0, 0,
IDATA   = 0,
IPRINT  = 1,
IOPT    = 2,
$END
```

continued

Sample Output from the Program

0.3-M TCT STREAMLINED WALL TEST SECTION
RIGHT SIDEWALL BOUNDARY-LAYER MEASUREMENTS

TEST:200 RUN: 19 POINT: 6

AVERAGE MACH NUMBER	:	.7049
AVERAGE REYNOLDS NUMBER/FT	:	.295E+08
AVERAGE STAGNATION PRESSURE PSIA	:	35.6035
AVERAGE STAGNATION TEMPERATURE K	:	129.7217
PRANDTL NUMBER	:	.7600
RATIO OF SPECIFIC HEATS	:	1.4000
RECOVERY FACTOR	:	.9126
REFERENCE LENGTH SCALE L	:	1.0000
VALUE OF N IN POWER LAW PROFILE	:	7.9196
BOUNDARY-LAYER THICKNESS DELTA/L	:	.5642
DISPLACEMENT THICKNESS DELTASTAR/L	:	.0713
MOMENTUM THICKNESS THETA/L	:	.0488
SHAPE FACTOR	:	1.4595

A: DISPLACEMENT AND MOMENTUM THICKNESS REFER TO COMPRESSIBLE VALUES.

B: DISPLACEMENT AND MOMENTUM THICKNESS CALCULATED USING POWER-LAW FROM WALL TO FIRST TUBE.

continued

NO.	Y/L	M/MINF	P/PT	T/TINF	U/UINF
1	.0600	.7469	.8693	1.0374	.7608
2	.1000	.7935	.8884	1.0315	.8059
3	.1400	.8172	.8979	1.0287	.8288
4	.1800	.8489	.9141	1.0239	.8590
5	.2200	.8677	.9236	1.0211	.8767
6	.2600	.8991	.9405	1.0162	.9063
7	.3000	.9149	.9494	1.0137	.9211
8	.3400	.9372	.9625	1.0101	.9419
9	.3800	.9650	.9785	1.0057	.9678
10	.4200	.9684	.9804	1.0052	.9709
11	.4600	.9777	.9860	1.0037	.9795
12	.5000	.9868	.9917	1.0022	.9878
13	.5500	.9940	.9962	1.0010	.9945
14	.5900	.9932	.9957	1.0011	.9938

\- - - - -

Appendix D: Program listing

```
      PROGRAM CRYOTBL(INPUT,OUTPUT,TAPE5=INPUT,TAPE7,TAPE10)

C     * * * * * * * * * * * * * * * * * * * * * * * * * * *
C                 0.3-M TCT BOUNDARY-LAYER CALCULATION
C
C     THIS PROGRAM CALCULATES THE SIDEWALL BOUNDARY-LAYER
C     DISPLACEMENT THICKNESS, MOMENTUM THICKNESS AND THE
C     SHAPE PARAMETER FROM THE TOTAL HEAD PRESSURE MEASURE-
C     MENTS IN THE BOUNDARY-LAYER.

C     * * * * * * * * * * * * * * * * * * * * * * * * * * *
C
      REAL MBL,MBLBMIN,MINF,L,MAV

      DIMENSION XX(30),YY(30),EPS(3),YM(30)
      DIMENSION W(30),RESID(30,1),SUM(1),A(30,1),B(30,1),WK(30,2)
      DIMENSION Y(30),PBL(30),PS(30),ARRAY(1000)
      DIMENSION P1(30),P2(30),P3(30),P4(30),P5(30)
      DIMENSION PTINF(30),TTINF(30),MBL(30)
      DIMENSION PBLNON(30),MBLBMIN(30),UBLBUIN(30),TBLBTIN(30)
      DIMENSION U(30),T(30),F(30),G(30),IBL(3)
      DIMENSION P(30),YN(30),PN(30),M(30)
      DIMENSION IORF1(30),PTC(30),PRT(30)
      DIMENSION HEADING(10)

      NAMELIST /PARAM/ NTEST,NRUN,NPOINT,NBL,NPORT,NBLSCAN,NPTINF,
     $NPTC,NPRT,ICOMP,IPLOT,Y,L,ISKIP,IBL,IDATA,IPRINT,IOPT

  600 CONTINUE
C     SET DEFAULT VALUES OF THE NAMELIST PARAMETERS

      IDATA   =     0
      NTEST   =   003
      NRUN    =   005
      NPOINT  =   006
      NBL     =   015
      NPORT   =   010
      NBLSCAN =   117
      NPRT   =    217
      NPTINF  =   107
      NPTC    =   108
      ICOMP   =   001
      IPLOT   =   001
      Y( 1)   = 0.024
      Y( 2)   = 0.060
      Y( 3)   = 0.100
      Y( 4)   = 0.140
      Y( 5)   = 0.180
      Y( 6)   = 0.220
      Y( 7)   = 0.260
      Y( 8)   = 0.300
      Y( 9)   = 0.340
```

```
      Y(10)    = 0.380
      Y(11)    = 0.420
      Y(12)    = 0.460
      Y(13)    = 0.500
      Y(14)    = 0.550
      Y(15)    = 0.590
      L        = 1.000
      ISKIP     =    1
      IPRINT    =    0
      IBL(1)    =    1
      IBL(2)    =    0
      IBL(3)    =    0
      IOPT      =    1

C     SET ERROR FLAG TO ZERO
      IERR = 0


C     READ INPUT DATA FROM NAMELIST

  140 READ(5, PARAM)
      IF(EOF(5)) 120, 130
  120 STOP 120
  130 CONTINUE
      IF(IDATA .EQ. 0) PRINT PARAM

C     STORE Y VALUES
      DO 1600 II = 1,NBL
      YM(II) = Y(II)
 1600 CONTINUE
      ENBL = NBL
      ISKP = ISKIP



      IF(IDATA .EQ. 0) GO TO 700

C     READ TITLE CARD
      READ 650, (HEADING(I),I=1,8)

C     READ INPUT DATA
      DO 710 I=1, NBL
      READ 720, PTINF(I),TTINF(I),Y(I),PBL(I),PTC(I)
      PRINT 720, PTINF(I),TTINF(I),Y(I),PBL(I),PTC(I)
  720 FORMAT(5F10.6)
  710 CONTINUE
      PRINT 660, (HEADING(I),I=1,8)
      GO TO 15

C     TUNNEL AND BOUNDARY-LAYER PARAMETERS ARE READ FROM EU TAPES

  700 CONTINUE
```

```
C     FIRST TWO FRAMES CONTAIN NON-DAT, INFOR SUCH AS UNITS AND LABELS
      DO 1000 I1=1,2
      READ(7)KEY,N,(ARRAY(I),I=1,N)
 1000 CONTINUE

C     SET REQUIRED INPUT DATA TO VARIABLES USED IN THE PROGRAM
      NSVO = NBLSCAN


    2 READ 4,  IT, IR, IP
      IF(IR. EQ. 999) GO TO 110

C     READ THE TUNNEL VARIABLES FROM THE EU TAPE

    5 READ(7)KEY,N,(ARRAY(I),I=1,N)
      IF(EOF(7))2,6
    6 CONTINUE

C     SELECT THE REQUIRED PARAMETERS

      IF (IP. NE. 0) GO TO 1500
      IF(ARRAY(NRUN) .GT. IR)  GO TO 2

 1500 CONTINUE
      ITEST=ARRAY(NTEST)
      IRUN=ARRAY(NRUN)
      IPOINT=ARRAY(NPOINT)
      IF(IRUN .EQ. 0) GO TO 7
      IF(IRUN .LT. IR) GO TO 5
      IF(IP .EQ. 0) GO TO 7
      IF(IPOINT .NE. IP) GO TO 5
    7 IPORT=ARRAY(NPORT)
      IF(IPORT .EQ. 0) GO TO 5
      IF(IPORT .GE. 16) GO TO 5
      IORF1(IPORT)=ARRAY(NPORT)
      PBL(IPORT)=ARRAY(NSVO)
      PTINF(IPORT)=ARRAY(NPTINF)
      PTC(IPORT)=ARRAY(NPTC)
      TTINF(IPORT)=ARRAY(NPRT)

      IF (IPORT.EQ.1) GO TO 10
      IF (IPORT.EQ.NBL) GO TO 15
      GO TO 5

   10 PRINT 40
      PRINT 100,ITEST,IRUN,IPOINT

   13 GO TO 5
   15 CONTINUE

      IF(ISKIP .NE. 0)
```

```
     $CALL SKIP(NBL,ISKIP,IBL,Y,PBL,PTINF,TTINF,PTC,IPRINT,IERR)
      SIGMA=0.76
      GAMMA=1.4
      N=NBL

C     CALCULATE BOUNDARY-LAYER PROFILES

      CALL BLPROFL(NBL,Y,PBL,PTC,PTINF,TTINF,L,GAMMA,SIGMA,
     $YN,PBLNON,MBLBMIN,UBLBUIN,TBLBTIN,MAV,PTINFAV,TTINFAV,
     $REYAV,REC,IPRINT,IERR)
      IF(IERR .NE. 0) GO TO 600

C     FIT A POWER-LAW CURVE TO THE VELOCITY PROFILE

      CALL LSFIT(NBL,YN,UBLBUIN,L,IPRINT,EN,DELTA,IERR)
      IF(IERR .NE. 0) GO TO 600

C     CALCULATE DISPLACEMENT THICKNESS AND MOMENTUM THICKNESS

      CALL THIKNES(NBL,YN,UBLBUIN,TBLBTIN,L,ICOMP,DELTA,EN,IOPT
     $,MAV,REC,GAMMA,DELS,THETA,IPRINT,IERR)


C     PRINT FINAL RESULTS
      PRINT 500, MAV
      PRINT 501, REYAV/3.2808
      PRINT 502, PTINFAV
      PRINT 503, TTINFAV
      PRINT 504, SIGMA
      PRINT 505, GAMMA
      PRINT 506, REC
      PRINT 507, L
      PRINT 508, EN
      PRINT 509, DELTA/L
      PRINT 510, DELS
      PRINT 511, THETA
      PRINT 512, DELS/THETA

      IF(ICOMP .EQ. 0) PRINT 521
      IF(ICOMP .NE. 0) PRINT 522
      IF(IOPT  .EQ. 1) PRINT 523
      IF(IOPT  .EQ. 2) PRINT 524

      PRINT 530

      DO 550  I = 1, NBL
      PRINT 531, I,YN(I),MBLBMIN(I),PBLNON(I),TBLBTIN(I),UBLBUIN(I)
  550 CONTINUE

C     WRITE PLOT FILE
      IF(IPLOT .EQ. 0) GO TO 570
```

```
      WRITE(10, 533) NBL, EN
      IF(IDATA .EQ. 0) WRITE(10, 534) ITEST,IRUN,IPOINT
      IF(IDATA .NE. 0) WRITE(10, 535) HEADING(1),HEADING(2)

      DO 560  I = 1, NBL
      WRITE(10, 536) Y(I)/DELTA, UBLBUIN(I)
  560 CONTINUE


  570 PRINT 532
      IF(IDATA .NE. 0) GO TO 600
      NBL = ENBL
      ISKIP = ISKP
C     RESET Y VALUES
      DO 1700 II = 1,NBL
      Y(II) = YM(II)
 1700 CONTINUE
      GO TO 5


  500 FORMAT(/10X,*AVERAGE MACH NUMBER                        :*, F10.4)
  501 FORMAT( 10X,*AVERAGE REYNOLDS NUMBER/FT                 :*, E10.3)
  502 FORMAT( 10X,*AVERAGE STAGNATION PRESSURE   PSIA         :*, F10.4)
  503 FORMAT( 10X,*AVERAGE STAGNATION TEMPERATURE  K          :*, F10.4)
  504 FORMAT( 10X,*PRANDTL NUMBER                             :*, F10.4)
  505 FORMAT( 10X,*RATIO OF SPECIFIC HEATS                    :*, F10.4)
  506 FORMAT( 10X,*RECOVERY FACTOR                            :*, F10.4)
  507 FORMAT( 10X,*REFERENCE LENGTH SCALE   L                 :*, F10.4)
  508 FORMAT(/10X,*VALUE OF N IN POWER LAW PROFILE            :*, F10.4)
  509 FORMAT( 10X,*BOUNDARY-LAYER THICKNESS DELTA/L           :*, F10.4)
  510 FORMAT(/10X,*DISPLACEMENT THICKNESS DELTASTAR/L         :*, F10.4)
  511 FORMAT( 10X,*MOMENTUM THICKNESS THETA/L                 :*, F10.4)
  512 FORMAT( 10X,*SHAPE FACTOR                               :*, F10.4)

  521 FORMAT(//10X,*A: DISPLACEMENT AND MOMENTUM THICKNESS REFER*,
     $/10X,       *   TO INCOMPRESSIBLE VALUES.*)
  522 FORMAT(//10X,*A: DISPLACEMENT AND MOMENTUM THICKNESS REFER*,
     $/10X,       *   TO COMPRESSIBLE VALUES.*)
  523 FORMAT(10X,*B: DISPLACEMENT AND MOMENTUM THICKNESS CALCULATED*,
     $       /10X,*   USING POWER-LAW FOR WHOLE PROFILE.*)
  524 FORMAT(10X,*B: DISPLACEMENT AND MOMENTUM THICKNESS CALCULATED*,
     $       /10X,*   USING POWER-LAW FROM WALL TO FIRST TUBE.*)

  530 FORMAT(///10X,*NO.*,6X,*Y/L*,4X,*M/MINF*,6X,*P/PT*,4X,
     $*T/TINF*,3X,*U/UINF*,/)
  531 FORMAT(10X,I2,5F10.4)
  532 FORMAT(//30X,*-   -   -   -   -*)
  533 FORMAT(I5,F10.6)
  534 FORMAT(*17HU/UE *,3I4,*      Y/DELTA      *)
  535 FORMAT(*17H U/UE  *,2A10,*      Y/DELTA     *)
  536 FORMAT(2F10.6)
```

```
650 FORMAT(8A10)
660 FORMAT(1H,/10X,8A10,/)

  4 FORMAT(3I3)
 40 FORMAT(1H1/10X,*0.3-M TCT STREAMLINED WALL TEST SECTION*,
   $/10X,*RIGHT SIDEWALL BOUNDARY-LAYER MEASUREMENTS*)
100 FORMAT(/10X,*TEST: *,I3,5X,*RUN:*,I3,5X,*POINT:*,I3)
101 FORMAT(3I10)
110 CONTINUE
    GO TO 140
    STOP
    END
```

```
      SUBROUTINE SKIP(NBL,ISKIP,IBL,Y,PBL,PTINF,TTINF,PTC,IPRINT,IERR)

      DIMENSION IBL(3),Y(30),PBL(30),PTINF(30),TTINF(30),PTC(30)
      DIMENSION P1(30),P2(30),P3(30),P4(30),P5(30)

C     * * * * * * * * * * * * * * * * * * * * * *
C
C     NBL    : NUMBER OF TUBES IN BOUNDARY-LAYER RAKE
C     ISKIP  : NUMBER OF POINTS TO BE SKIPPED (MAXIMUM 3)
C     IBL(3) : PORT NUMBERS TO BE SKIPPED
C     Y      : DISTANCE OF TUBES FROM WALL
C     PBL    : TOTAL PRESSURE IN THE BOUNDARY-LAYER
C     PTINF  : FREE-STREAM STAGNATION PRESSURE
C     TTINF  : FREE-STREAM STAGNATION TEMPERATURE
C     PTC    : FREE-STREAM STATIC PRESSURE
C     IPRINT : NOT USED
C     IERR   : NOT USED
C
C     NOTE   : ON RETURN, NEW VALUES OF NBL,Y,PBL,PTINF,TTINF,PTC
C              WILL BE AVAILABLE.
C
C     * * * * * * * * * * * * * * * * * * * * * *

      JS = 0
  300 DO 100 I=1,NBL
      DO 200 J=1,ISKIP
      IF(I .EQ. IBL(J)) GO TO 150
  200 CONTINUE
      GO TO 250

  150 JS = JS + 1
      GO TO 100


  250 I1= I - JS
      P1(I1)= PBL(I)
      P2(I1)= PTINF(I)
      P3(I1)= TTINF(I)
      P4(I1)= PTC(I)
      P5(I1)= Y(I)
  100 CONTINUE

      NBL= NBL - ISKIP

      DO 400 I=1, NBL
      PBL(I)   = P1(I)
      PTINF(I) = P2(I)
      TTINF(I) = P3(I)
      PTC(I)   = P4(I)
      Y(I)     = P5(I)
  400 CONTINUE
```

```
      IF(IPRINT .EQ. 0) RETURN
      PRINT 500
      DO 550 I=1, NBL
      PRINT 600, I,Y(I),PTINF(I),TTINF(I),PTC(I),PBL(I)
  550 CONTINUE

  500 FORMAT(/10X,*REARRANGED VALUES OF THE PARAMETERS Y,PTINF,*,
     $/,        *TTINF,PTC,PBL BY SUBROUTINE SKIP ARE AS FOLLOWS*,/)
  600 FORMAT(I10,5F12.6)
      RETURN
      END
```

```
      SUBROUTINE LSFIT(N,Y,U,L,IPRINT,EN,DELTA,IERR)

C     * * * * * * * * * * * * * * * * * * * * * *
C
C     THIS SUBROUTINE FIT A POWER-LAW CURVE TO THE VELOCITY PROFILE TO
C     DETERMINE THE CONSTANTS EN AND DELTA
C
C     N        :  NUMBER OF POINTS IN VELOCITY PROFILE
C     Y(N)     :  VALUES OF Y
C     U(N)     :  VALUES OF NON-DIMENSIONALIZED VELOCITY
C     L        :  LENGTH SCALE
C     IPRINT   :  NOT,EQUAL TO ZERO, PRINTS OUTPUT PARAMETERS
C     IERR     :  EQUAL TO ZERO, NORMAL RETURN
C              :  NOT EQUAL TO ZERO, ERROR IN CURVE FITTING
C
C     EN       :  INDEX IN POWER-LAW  VELOCITY PROFILE
C     C2       :  CONSTANT TERM       DELTA = L/EXP(C2/C1)
C
C                 THE CURVE FIT IS OF THE TYPE
C                 LOGN(U/UINF) = C1 * LOGN(Y/L) + C2
C
C     * * * * * * * * * * * * * * * * * * * * * *

      REAL L
      DIMENSION Y(30),U(30)
      DIMENSION XX(30),YY(30)
      DIMENSION W(30),RESID(30,1),SUM(1),A(30,1),B(30,1),WK(30,2)

      DO 100 I = 1, N
      XX(I) = ALOG(Y(I))
      YY(I) = ALOG(U(I))
      W(I)  = 1
  100 CONTINUE

C     SET PARAMETERS FOR LEAST SQUARE FIT

      NMAX  = 30
      N     = N
      LL    = 1
      MMAX  = 4
      M     = 2

C     LEAST SQUARE FIT FOR N READINGS

      CALL LSQPOL(NMAX,N,XX,LL,YY,W,MMAX,M, RESID,SUM,A,B,WK,IERR)
      IF(IERR .NE. 0) PRINT 500
  500 FORMAT(/10X,*ERROR IN SUBROUTINE LSQPOL:  CHECK VELOCITY DATA*)
      IF(IERR .NE. 0) RETURN
```

```
C     DETERMINE EN AND DELTA
      EN    = 1./B(2,1)
      DELTA = L/EXP(B(1,1)/B(2,1))

      IF (IPRINT .EQ. 0) GO TO 300

C     PRINT INPUT TO AND OUTPUT FROM SUBROUTINE LSQPOL

      PRINT 1000, NMAX,N,LL,MMAX,M
      DO 200 I= 1, N
      PRINT 2000, I,W(I),Y(I),U(I),XX(I),YY(I),RESID(I,1)
  200 CONTINUE

      PRINT 2020, IERR, EN, DELTA, SUM(1)

  300 CONTINUE

 1000 FORMAT(10X,5I5)
 2000 FORMAT (5X,I5,6F9.4)
 2020 FORMAT(10X,*IERR =*,I2,5X,*EN =*,F8.4,5X,*DELTA=*,F8.4,
     $5X,*SUM(1)=*,F8.4)
 2030 FORMAT(/10X,F10.6,/10X,5E10.4)
      RETURN
      END
```

```
      SUBROUTINE BLPROFL(N,Y,PBL,PS,PTINF,TTINF,L,GAMMA,SIGMA,
     $YN,PBLNON,MBLBMIN,UBLBUIN,TBLBTIN,MAV,PTINFAV,TTINFAV,
     $REYAV,REC,IPRINT,IERR)
      REAL MBL,MBLBMIN,MINF,L,MAV
      DIMENSION Y(30),PBL(30),PS(30)
      DIMENSION PTINF(30),TTINF(30),MINF(30),MBL(30)
      DIMENSION YN(30),PBLNON(30),MBLBMIN(30),UBLBUIN(30),TBLBTIN(30)

C     * * * * * * * * * * * * * * * * * * * * * 
C
C     N       : NUMBER OF TUBES IN THE BOUNDARY-LAYER RAKE
C     Y       : PROBE HEIGHTS FROM THE WALL
C     PBL     : TOTAL PRESSURE IN THE BOUNDARY-LAYER
C     PS      : STATIC PRESSURE (FREE-STREAM)
C     PTINF   : TUNNEL STAGNATION PRESSURE
C     TTINF   : TUNNEL STAGNATION TEMPERATURE
C     L       : LENGTH SCALE FOR NON-DIMENSIONALISING Y
C     GAMMA   : RATIO OF SPECIFIC HEATS
C     SIGMA   : PRANDTL NUMBER
C
C     YN      : VALUES OF Y NORMALISED BY L
C     PBLNON  : VALUES OF PBL NORMALISED BY PTINF
C     MBLBMIN: RATIO OF MACH NUMBER IN BOUNDRY-LAYER TO FREE-STREAM
C     UBLBUIN: RATIO OF VELOCITY IN BOUNDARY-LAYER TO FREE-STREAM
C     TBLBTIN: RATIO OF TEMPERATURE IN BOUNDARY-LAYER TO FREE-STREAM
C     MAV     : AVERAGE MACH NUMBER
C     PTINFAV: AVERAGE STAGNATION PRESSURE
C     TTINFAV: AVERAGE STAGNATION TEMPERATURE
C     REYAV   : AVERAGE REYNOLDS NUMBER/METER
C     REC     : RECOVERY FACTOR
C     IPRINT :
C     IERR    :
C
C     * * * * * * * * * * * * * * * * * * * * * *

C     RECOVERY FACTOR

      R = SIGMA**(1./3.)
      REC = SIGMA**(1./3.)

C     CONSTANTS

      G = GAMMA
      G1= (G-1.)/G
      G2= 1./G
      G3= R*(G-1.)/2.
```

```
C     CALCULATE MACH NUMBERS IN BOUNDARY-LAYER AND FREESTREAM, AND RATIO

      DO 100 I = 1, N
      X1  = (PBL(I)/PS(I))**G1 -1.
      X2 =(PTINF(I)/PS(I))**G1 -1.
      IF(X1 .LT. 0.  .OR. X2 .LT. 0.) GO TO 1000
      X1 = (2./(G-1.))*X1
      X2 = (2./(G-1.))*X2
      MBL(I)  = SQRT(X1)
      MINF(I) = SQRT(X2)
      MBLBMIN(I) = MBL(I)/MINF(I)
      PBLNON(I)  = PBL(I)/PTINF(I)
  100 CONTINUE

C     CALCULATE RATIO OF TEMPERATURE IN BOUNDARY-LAYER TO FREESTREAM

      DO 200 I = 1, N
      TBLBTIN(I)= (1. + G3* (MINF(I)**2.) ) / (1. + G3* (MBL(I)**2.))
  200 CONTINUE

C     CALCULATE RATIO OF VELOCITY IN BOUNDARY-LAYER TO FREESTREAM

      DO 300 I = 1, N
      UBLBUIN(I) = MBLBMIN(I) * SQRT(TBLBTIN(I))
  300 CONTINUE

C     NON-DIMENSIOALIZE RAKE PROBE HEIGHTS TO LENGTH SCALE L

      DO 400 I = 1, N
      YN(I) = Y(I)/L
  400 CONTINUE

C     CALCULATE AVERAGE QUANTITIES

      SUMM = 0.
      SUMP = 0.
      SUMT = 0.

      DO 500  I = 1, N
      SUMM = SUMM + MINF(I)
      SUMP = SUMP + PTINF(I)
      SUMT = SUMT + TTINF(I)
  500 CONTINUE

      MAV     = SUMM/N
      PTINFAV = SUMP/N
      TTINFAV = SUMT/N

      IF(IPRINT .EQ. 0) GO TO 1010
      PRINT 1020
      DO 1030  I =1, N
      PRINT 1040, I, MINF(I),MBL(I),UBLBUIN(I),TBLBTIN(I)
```

```
 1030 CONTINUE
 1020 FORMAT(/1X,*      I        MINF      MBL        U/UINF     T/TINF*)
 1040 FORMAT(1X,I5,4F10.6)
C      CALCULATE AVERAGE REYNOLDS NUMBER/METER

 1010 CONTINUE
      C     = 1. + ((G - 1.)/2.) * (MAV**2.)
      A1    = PTINFAV/(TTINFAV**1.4)
      A2    = MAV/(C**2.1)
      A3    = (64314./14.696)*(10.**6.)
      Z     = COMP(PTINFAV,TTINFAV)
      REYAV= A3*A2*A1*Z
      IF(IPRINT .NE. 0) PRINT 1050, MAV,PTINFAV,TTINFAV,Z,REYAV
 1050 FORMAT(/10X,*MAV, PTINFAV, TTINFAV, Z, REYAV:  *, /10X,
     $4F10.4,E14.4)



      RETURN
 1000 PRINT 1100
 1100 FORMAT(/10X,*ERROR IN CALCULATING BOUNDARY-LAYER VELOCITY*,
     $* PROFILE:  CHECK INPUT PRESSURE DATA*)
      RETURN
      END
```

```
      SUBROUTINE THIKNES(N,Y,U,T,L,ICOMP,DELTA,EN,IOPT
     $,MINF,REC,GAMMA,DELS,THETA,IPRINT,IERR)
      REAL MINF, L
      DIMENSION Y(30),U(30),T(30),F(30),G(30)
C     * * * * * * * * * * * * * * * * * * * * * *
C
C     N       :  NUMBER OF POINTS IN THE BOUNDARY-LAYER
C     Y       :  RAKE PROBE HEIGHTS
C     U       :  NON-DIMENSIONALISED VELOCITY PROFILE AT N POINTS
C     T       :  NON-DIMENSIONALISED TEMPERATURE PROFILE AT N POINTS
C     L       :  LENGTH SCALE
C     ICOMP   :  = 0, CALCULATES INCOMPRESSIBLE VALUES
C             :  / 0, CALCULATES COMPRESSIBLE VALUES
C     DELTA   :  BOUNDAR-LAYER THICKNESS
C     EN      :  CONSTANT IN POWER-LAW VELOCITY PROFILE
C     MINF    :  MACH NUMBER
C     REC     :  RECOVERY FACTOR
C     GAMMA   :  RATIO OF SPECIFIC HEATS
C     IPRINT  :  NOT = ZERO, PRINTS INTERMEDIATE RESULTS
C     IOPT    :  = 1, DELTASTAR CALCULATED USING EN AND DELTA
C                     FOR THE ENTIRE PROFILE
C     IOPT    :  = 2, DELTASTAR CALCULATED USING EN AND DELTA
C                     ONLY UPTO FIRST TUBE, AND USING TRAPEZOIDAL
C                     RULE FOR OTHER POINTS.
C     IERR    :  NOT USED

C     DELS    :  DISPLACEMENT THICKNESS / L
C     THETA   :  MOMENTUM THICKNESS / L

C
C     * * * * * * * * * * * * * * * * * * * * * *

      NBL = N
      IF(IOPT .EQ. 1) GO TO 680


C     CALCULATE THE VELOCITY/MASS DEFECT

      DO 100 I= 1, N
      IF(ICOMP .NE. 0) F(I) = 1. - U(I)/T(I)
      IF(ICOMP .NE. 0) G(I) = (U(I)/T(I))*(1. - U(I))
      IF(ICOMP .EQ. 0) F(I) = 1. - U(I)
      IF(ICOMP .EQ. 0) G(I) = U(I)*(1. - U(I))
  100 CONTINUE

C     INTEGRATE FROM TUBE(1) TO TUBE(N)

      IF(IPRINT .NE. 0 .AND. ICOMP .NE. 0) PRINT 500
      IF(IPRINT .NE. 0 .AND. ICOMP .EQ. 0) PRINT 510
```

```
 500 FORMAT(/1X,*   I         Y         U       T       (1 - U/T)      SUM*,
    $*   (U/T)(1-U)  SUM*)
 510 FORMAT(/1X,*   I         Y         U       T      (1-U)       SUM*,
    $*     U(1-U)     SUM*)


     SUMI = 0.
     SUMG = 0.
     DO 200 I= 2, N
     IF(I .EQ. 2  .AND. IPRINT. NE. 0)
    $PRINT 600, (I-1),Y(1),U(1),T(1),F(1),SUMI,G(I),SUMG
     DY   = Y(I) -Y(I-1)
     SUMI= SUMI + (F(I)+F(I-1))*(DY/2.)
     SUMG= SUMG + (G(I)+G(I-1))*(DY/2.)
     IF(IPRINT .NE. 0)
    $PRINT 600, I,Y(I),U(I),T(I),F(I),SUMI,G(I),SUMG
 600 FORMAT(1X,I2,7F10.4)

 200 CONTINUE

     DELS = SUMI
     THETA= SUMG

C    CONTRIBUTION FROM WALL TO FIRST TUBE

     A = 0.
     B = Y(1)/DELTA
     CALL INCRS(MINF,REC,GAMMA,A,B,DELTA,L,EN,DELS1,THETA1,
    $IPRINT,IERR1,IERR2)
     IF(IERR1 .LE. 2 .AND. IERR2 .LE. 2) GO TO 630
     PRINT 610, IERR1, IERR2
 610 FORMAT(/10X,*ERROR IN SUBROUTINE CADRE WHILE CALCULATING*,
    $10X,*CONTRIBUTION FROM WALL TO FIRST TUBE:*,
    $10X,*ERROR NUMBERS :*,2I5)
     RETURN
 630 DELS     = DELS + DELS1 *(DELTA/L)
     THETA    = THETA+ THETA1*(DELTA/L)
     IF(IPRINT .NE. 0)
    $PRINT 660, DELS1,DELS    ,THETA1,THETA
 660 FORMAT(/10X,*DELS1 DELS     THETA1 THETA*,4F10.6)
     IF(DELTA .LE. Y(NBL)) RETURN


C    CONTRIBUTION FROM LAST TUBE TO  DELTA

     A = Y(NBL)/L
     B = 1.0
     CALL INCRS(MINF,REC,GAMMA,A,B,DELTA,L,EN,DELS2,THETA2,
    $IPRINT,IERR1,IERR2)
     IF(IERR1 .LE. 2 .AND. IERR2 .LE. 2) GO TO 650
     PRINT 640, IERR1, IERR2
```

```
  640 FORMAT(/10X,*ERROR IN SUBROUTINE CADRE WHILE CALCULATING*,
     $10X,*CONTRIBUTION FROM LAST TUBE TO DELTA:*,
     $10X,*ERROR NUMBERS :*,2I5)
      RETURN
  650 DELS     = DELS + DELS2 *(DELTA/L)
      THETA    = THETA+ THETA2*(DELTA/L)
      IF(IPRINT .NE. 0)
     $PRINT 670, DELS2,THETA2
  670 FORMAT(/10X,*DELS2 THETA2:*,2F10.6)
      IF(DELTA .LE. Y(NBL)) RETURN

C     IOPT = 2: CALCULATE DELTASTAR AND THETA FROM Y=0 TO DELTA

  680 CONTINUE
      A = 0.0
      B = 1.0
      CALL INCRS(MINF,REC,GAMMA,A,B,DELTA,L,EN,DELS ,THETA ,
     $IPRINT,IERR1,IERR2)

      IF(IERR1 .LE. 2 .AND. IERR2 .LE. 2) GO TO 700
      PRINT 690, IERR1,IERR2

  690 FORMAT(/10X,*ERROR IN SUBROUTINE CADRE WHILE CALCULATING*,
     $/10X,* FOR IOPT = 1 IN THE NAMELIST*,
     $/10X,*ERROR NUMBER: *,2I5)

  700 DELS = DELS * (DELTA/L)
      THETA= THETA* (DELTA/L)
      IF(IPRINT .NE. 0) PRINT 710, DELS,THETA
  710 FORMAT(/10X,*DELS3 THETA3:  *,2F10.6)


      RETURN
      END
```

```
      SUBROUTINE INCRS(MINF,REC,GAMMA,A,B,DELTA,L,EN,DELS1,THETA1,
     $IPRINT,IERR1,IERR2)

      REAL MINF,L
      DIMENSION EPS(3)
      COMMON /AA/ C,ENP
      EXTERNAL FDEL, FTHETA

C     * * * * * * * * * * * * * * * * * * * * * * *
C
C     MINF    :  FREE-STREAM MACH NUMBER
C     REC     :  RECOVERY FACTOR
C     GAMMA   :  RATIO OF SPECIFIC HEATS
C     A       :  LOWER LIMIT OF INTEGRATION
C     B       :  UPPER LIMIT OF INTEGRATION
C     EN      :  INDEX IN POWER LAW (U/UE)=(Y/D)**(1/N)
C     L       :  LENGTH SCALE
C     DELTA   :  BOUNDARY-LAYER THICKNESS (BY POWER-LAW FIT)
C
C     DELS1   :  VALUE OF DELSTAR/L FROM WALL TO NEAREST TUBE
C     THETA1  :  VALUE OF MOMENTM/L FROM WALL TO NEAREST TUBE
C     IERR1   :  NOT = 0, ERROR INTEGRATING DISPLACEMENT THICHNESS
C     IERR2   :  NOT = 0, ERROR INTEGRATING MOMENTUM THICKNESS
C
C     * * * * * * * * * * * * * * * * * * * * * * *

      C       = (REC * (GAMMA -1.)/2.) * (MINF*MINF)
      ENP = EN

      IF(IPRINT .NE. 0) PRINT 100, MINF,REC,GAMMA,A,B,ENP,L,DELTA
  100 FORMAT(8F10.6)

      EPS(1)  = 1.0E-10
      EPS(2)  = 1.0E-08
      ITEXT   = 0

      CALL CADRE(A,B,FDEL,EPS,ITEXT,SUM,IERR)
      DELS1   = SUM
      IERR1   = IERR
      IF(IPRINT .NE. 0) PRINT 200, SUM, IERR
  200 FORMAT(10X,F10.6,I5)


      CALL CADRE(A,B,FTHETA,EPS,ITEXT,SUM,IERR)
      THETA1  = SUM
      IERR2   = IERR
      IF(IPRINT .NE. 0) PRINT 200, SUM, IERR

      RETURN
      END
```

```
      FUNCTION FDEL(YD)
      COMMON /AA/ C,ENP

      IF(YD .NE. 0.) U     = YD**(1./ENP)
      IF(YD .EQ. 0.) U     = 0.
      T     = 1. + C* (1. - U*U)
      FDEL = (1. - U/T)
C     PRINT 100, YD,U,T,FDEL,C,ENP
  100 FORMAT(10X,6F10.6)

      RETURN
      END

      FUNCTION FTHETA(YD)
      COMMON /AA/ C,ENP

      IF(YD .NE. 0.) U     = YD**(1./ENP)
      IF(YD .EQ. 0.) U     = 0.
      T     = 1. + C* (1. - U*U)
      FTHETA= (U/T)*(1.-U)

      RETURN
      END
      FUNCTION COMP(P,T)

C     CALCULATES COMPRESSIBILITY FACTOR Z  FOR  GIVEN
C     PRESSURE P(PSIA) AND TEMPERATURE  T(DEG K)

      P=P/14.7
      B0= 1.37
      B1= -8.773E-02
      B2= 4.703E-04
      B3= -1.386E-06
      B4= 1.462 E-09
      C0= 5.521
      C1= -1.986E-01
      C2= 7.817 E-04
      C3= -1.258E-06
      C4= 5.333E-10
      D=B0+B1*T +B2*T*T +B3*(T**3.) +B4*(T**4.)
      E=C0+C1*T +C2*T*T +C3*(T**3.) +C4*(T**4.)
      Z=1. -P* EXP(D) -(P*P)* EXP(E)
      COMP=Z
      P=P*14.7
      RETURN
      END
/
```



Figure 1: Sketch of 0.3-Meter Transonic Cryogenic Tunnel Boundary-Layer Rake.



Figure 2: Determination of edge of boundary-layer and constant 'n' in the power-law velocity profile ($U \alpha y^{1/n}$).



Figure 3: Typical plot of measured boundary-layer velocities and power-law fit.

1. Report No.	2. Government Accession No.	3. Recipient's Catalog No.
NASA CR-178241		

4. Title and Subtitle	5. Report Date
Calculation of Sidewall Boundary-Layer Parameters From Rake Measurements for the Langley 0.3-Meter Transonic Cryogenic Tunnel	February 1987
	6. Performing Organization Code
7. Author(s)	8. Performing Organization Report No.
A. V. Murthy	
	10. Work Unit No.
9. Performing Organization Name and Address	11. Contract or Grant No.
Vigyan Research Associates, Inc. 28 Research Road Hampton, VA 23666	NAS1-17919
	13. Type of Report and Period Covered
12. Sponsoring Agency Name and Address	Contractor Report
National Aeronautics and Space Administration Washington, DC 20546	14. Sponsoring Agency Code 505-61-01-02

15. Supplementary Notes

Technical Monitor: Raymond E. Mineck, Langley Research Center, Hampton, Virginia.

16. Abstract

Correction of airfoil data for sidewall boundary-layer effects requires a knowledge of the boundary-layer displacement thickness and the shape factor with the tunnel empty. To facilitate calculation of these quantities under various test conditions for the Langley 0.3-m Transonic Cryogenic Tunnel, a computer program has been written. This program reads the various tunnel parameters and the boundary-layer rake total head pressure measurements directly from the Engineering Unit tapes to calculate the required sidewall boundary-layer parameters. Details of the method along with the results for a sample case are presented.

17. Key Words (Suggested by Author(s))	18. Distribution Statement
Wind tunnels Boundary-layers	Unclassified - Unlimited Subject Category - 02

19. Security Classif. (of this report)	20. Security Classif. (of this page)	21. No. of Pages	22. Price
Unclassified	Unclassified	35	A03